
TUESDAY TIPS – SAS PROGRAMMING 

Weekly Tip for Nov. 3, 2020 

Your data dictionary contains some variables that can iterate - for example, the unknown # of COVID tests 
for an unknown # of infants. How do you manage documentation? 

A data dictionary for a file based on Electronic Medical Records (EMR) contains variables which represent an 
unknown number of COVID tests for an unknown number of infants – there is no way to know in advance 
how many iterations of this variable will exist in the actual data file. In addition, variables in this file may exist 
for three different groups (pregnant women, postpartum women, and infants), with PR, PP and IN prefixes, 
respectively. This Tuesday Tip demonstrates how to process such variables in a data dictionary to drive label 
(and value label) description creation for iterated (and other) labels  using SAS functions, as well as other 
utilities. 

 

Using PROC CONTENTS and ODS OUTPUT on an imported data dictionary (example shown above) and a data 
file from a health care entity, the position ODS OUTPUT object is created, and the column variable is 
standardized using the UPCASE function. 

************************************************************; 
*** Import Personal Data Dictionary one tab at a time    ***; 
************************************************************; 
  
%macro imptabs(tabn=1, tabnm=identifiers, intab=Identifiers, startrow=10, endcol=H); 
  

proc import dbms=xlsx out = temp datafile = " \file.xlsx" replace; 
      RANGE="&intab.$A&startrow.:&endcol.999"; 
      getnames=YES; 
run; 
  
. . . 
  
data labels&tabn.; 
   length label labelstr $ 300 variable_type $ 8; 
   set &tabnm (keep=variable_: pw_preg pw_pp inf 
         where=(variable_name ne '' and variable_description ne '')); 
     label=catx(": ","&tabnm.",variable_description); 
      labelstr=cats(variable_name,'="',label,'"'); 
  
     variable_length=length(variable_name); 
     length_flag=(variable_length+7 GT 32); 

     label variable_length="Length of Variable" 
             length_flag="Current Variable Length + 7 exceeds 32"; 
  
        /* find out the # of iterations within a variable name */ 
     iteration_flag=(indexc(variable_name,'#') gt 0); 
     iteration_count=countc(variable_name,'#'); 
  
     label iteration_flag="Binary: Variable iterations" 

iteration_count="# of iteration points within variable name"; 
run; 
  
%mend; 
  
%imptabs(tabn=1, tabnm=Identifiers, intab=Identifiers, startrow=4, endcol=H); 

First, the LENGTH function is used to calculate the length of “variable”. The length of variable names is 
limited to 32 columns, and the name an of iterated variable may exceed the limits. The data dictionary is a 
living document, and if any overlong variables that exist once prefixes and iterated counts are added to the 
base, the spelling is adjusted. Identification variables are exempt from prefixes. When a file is first processed, 
variables, with the exception of ID variables, have prefixes added, using the CATS function. Additionally, label 
strings are created by concatenating prefixes and existing variable descriptions using the CATX function. 

It is relatively simple to replace a single iterator, in this case, a #, in a variable name. It is more complicated to 
replace two or more iterators. SAS functions process one variable transformation at a time – that is, they 
stop after completing a single operation on a string. We look for # in a variable and flag using the INDEX 
function. The INDEX function returns the position of the first pound sign. We then use the SUBSTR function to 
replace the # using a do loop, outputting additional label records for each iteration. Please see the example 
below. We use the COUNTC function to discover how many #s exist in a variable name.  In practice, this is 
done in a multidimensional array with a dimension for each # sign in a variable, with the maximum number 
of possible iterations. You can use functions to discover the number of iterations needed as well in the actual 
data – including the REVERSE and ANYNUM functions – in the actual data. Additionally, the iteration numbers 
are added to the label strings using CAT functions. Multiple supplement label files are created until no more # 
appear in the variable names. 

data expand_labels; 
     length variable $ 32; 
     set labels0 (where=(index(variable,'#')>0)); 
  
     *get the first indexed # location; 
     loc=index(variable,'#'); 
  
     substr(variable,loc,1)="1"; 
        output; 
     substr(variable,loc,1)="2"; 
        output; 
run; 

Multiple supplement label files are created until no more # appear in the variable names.  The files are 
concatenated and sorted to create a file of labels for potentially thousands of variables, which is matched to 
the variables in the metadata for actual files. This file can be included when processing an actual data file. 

 

data tolabel; 
     file label1 lrecl=300; 

     set matchtest (where=(inlabels=1 and inpos=1)); 
     put labelstr; 
run; 

The same process of concatenation based on metadata elements is used to create macro calls to create a 
codebook, a range report and a “missingness” report. We hope you’ll have some fun iterations with functions 
with your metadata as well! 

 

This week’s tip was contributed by Louise Hadden. Louise Hadden is a Lead Programmer Analyst in the Health 
and Environment Division at Abt Associates Inc., specializing in analysis and reporting. 

 

Weekly Tip for Nov. 17, 2020 

We all make the occasional assumption about variable values when we code. This tip describes a useful 
technique for checking and documenting your assumptions when you make them. 

Consider the simple task of converting a numeric GENDER variable to a character SEX variable.  

 

The main shortcoming of this code is that it is assuming that 1 and 2 are the only possible values of GENDER. 
What happens if you have a GENDER value of 3, or 99, or missing? How do you defend against all possible 
unexpected values of GENDER without having to write 27,463 additional ELSE statements?  

The technique is called “custom warnings”. It involves writing a mere line of code which, if triggered, will 
result in a WARNING being written to your log. 

If the incoming data set has only values of 1 and 2 for GENDER, then you’ll have a clean log. But if any other 
value occurs (in this case 3), you’ll get a WARNING in your log to alert you to the fact that action needs to be 
taken. 

Notice that the word WARNING is broken up into two pieces in the ELSE statement. This is done so that the 
word WARNING will not appear in your log unless the PUT is triggered.  

A variation of this technique can also be applied to the macro language. Imagine you have a macro 
parameter &POS that must be non-missing. You can put in a check at the top of your macro. 

%if &pos eq %str() %then 
   %put %str(W)ARNING: POS is a required parameter; 

Notice that the details in the macro language are slightly different. Instead of splitting up the word WARNING 
using quotation marks, we instead split it up using a macro quoting function.  

A collection of over a dozen additional applications of custom warnings, collected from my coworkers, is 
available on GitHub. 

https://github.com/srosanba/sas-custom-warnings/wiki 

This week’s tip was contributed by Shane Rosanbalm. Shane is a Senior Biostatistician at Rho, Inc., with a 
particular interest in SAS macro efficiencies, as well as graphical presentations of data.  

 

Weekly Tip for Nov. 24, 2020 

What simple tips can you use to improve the readability of your SAS output?  

The SAS system option shown below returns a line instead of a page break between pages of output.  It 
makes the review of your LST file much more efficient, and I literally use this every time I'm looking at SAS 

https://github.com/srosanba/sas-custom-warnings/wiki

interactively. This is the single most-used tip I have ever encountered, courtesy of Joy Smith a number of 
years ago. 

************************************************************; 
OPTIONS FORMDLIM = '_'; 
************************************************************; 

For more information on the FORMDLIM system option, see the SAS documentation  

FORMDLIM= System Option 

 

Another favorite is the WIDTH=MIN option in PROC PRINT, to condense your output for better readability.  I 
use this enough that I've long since created a little macro that I call "Venita Print", or %VPR. I typically put 
this at the beginning of all my programs as well.  

 %MACRO vpr (dset, obs); 
    %if "&obs"="" %then %do;   
        Title "--- &dset. ---";  
        proc print data=&dset (obs=&obs) width=min;  
        run;  
        Title;  
    %end; 
    %else %do;  
        Title "--- &dset. ---";  
        proc print data=&dset width=min;  
        run;  
        Title;  
    %end; 
%MEND vpr; 

For more information on the WIDTH= PROC PRINT OPTION, see the SAS  

WIDTH= Option for PRINT Procedure  

Happy SASsing!   

This week’s tip was contributed by Venita DePuy. Venita is the owner of Bowden Analytics and can be 
contacted at bowden.analytics@gmail.com.  

 

Weekly Tip for Dec. 8, 2020 

Removing Duplicate Texts in a String   

Sometimes we need to remove duplicate texts in a string. From my point of view, using traditional SAS 
functions such as SCAN and SUBSTR to solve this issue might be confusing and laborious. Here is a regular 
expression solution. 
  

https://documentation.sas.com/?cdcId=pgmsascdc&cdcVersion=9.4_3.5&docsetId=lesysoptsref&docsetTarget=n0vr78nr6afoxzn158j8cj5g224s.htm&locale=en
https://documentation.sas.com/?cdcId=pgmsascdc&cdcVersion=9.4_3.5&docsetId=proc&docsetTarget=n17dcq1elcvpvkn1pkecj41cva6j.htm&locale=en#n1fdxyvx5596zun1ef52j6x4fxwu
mailto:bowden.analytics@gmail.com

data _null_; 
       infile cards truncover; 
       input STRING $32767.; 
       REX1=prxparse('s/([a-z].+?\.\s+)(.*?)(\1+)/\2\3/i'); 
       REX2=prxparse('/([a-z].+?\.\s+)(.*?)(\1+)/i'); 
       do i=1 to 100; 
       STRING=prxchange(REX1, -1, compbl(STRING)); 
       if not prxmatch(REX2, compbl(STRING)) then leave; 
       end; 
       put STRING=; 
cards; 
a. The cow jumps over the moon. b. The chicken crossed the road. c. The quick brown 
fox jumped over the lazy dog. a. The cow jumps over the moon.  
b. The chicken crossed the road. a. The cow jumps over the moon. b. The chicken 
crossed the road. c. The quick brown fox jumped over the lazy dog. 
a. The cow jumps over the moon. a. The cow jumps over the moon. b. The chicken 
crossed the road. b. The chicken crossed the road. c. The quick brown fox jumped 
over the lazy dog. c. The quick brown fox jumped over the lazy dog. 
a. The cows jump over the moon. a. The cows jump over the moon. b. The chickens 
crossed the road. b. The chickens crossed the road. c. The quick brown foxes jumped 
over the lazy dog. c. The quick brown foxes jumped over the lazy dog. 
a. The cow jumps over the moon. b. The chicken crossed the road.  c. The quick 
brown fox jumped over the lazy dog. a. The cow jumps over the moon.  b. The chicken 
crossed the road. c. The quick brown fox jumped over the lazy dog. 
; 
run; 

 

Regular expression visualization by Regexper: 

 
Here’s a brief explanation of the regular expression. “[a-z]” matched a single lower-case letter. “.+?” matches 
any characters as few times as possible. “.” matches exactly a period character. “\s+” matches exactly a space 
as many times as possible. “.*?” matches any characters as few times as possible. “\1+” matches the first 
capturing group as many times as possible. 

Note that if the repeated time value is greater than 100, you need to increase the stopping value in DO loop 
accordingly. I think this scenario rarely happens. If you want to remove duplicate words instead of sentences, 
you need to adjust the expression. For example: 

data _null_; 
    STRING='cow chicken fox cow chicken fox cows chickens foxes'; 
    REX1=prxparse('s/(\b\w+\b)(.*?)(\b\1+\b)/\2\3/i'); 
    REX2=prxparse('/(\b\w+\b)(.*?)(\b\1+\b)/i'); 
    do i=1 to 100; 
        STRING=prxchange(REX1, -1, compbl(STRING)); 
        if not prxmatch(REX2, compbl(STRING)) then leave; 
    end; 
    put STRING=; 
run; 

Regular expression visualization by Regexper: 

 
“\b” matches a word boundary. “\w+” matches any word character (equal to [a-zA-Z0-9_]) as many times as 
possible. 

This week’s tip was contributed by Allen Zeng / 曾宪华. Xianhua Zeng is a principal statistical programmer at 
PAREXEL with 7+ years of experience in clinical trials industry.  More information on this topic may be viewed 
at Removing Duplicate Texts in a String. 

 

Weekly Tip for Dec. 15, 2020 

Compare & Warn   

Sick and tired of having to look at your output to determine whether or not your PROC COMPARE ran 
cleanly!? Unburden thyself, dear reader.  

SAS® generates an automatic macro variable called SYSINFO after PROC COMPARE has run. This macro 
variable contains information about the results obtained by the procedure. The macro %CompareWarn 
(https://github.com/srosanba/sas-comparewarn) has been created to interrogate SYSINFO and write 
messages to the log about any issues that were detected. 

The code is quite simple; you put the macro call immediately after your PROC COMPARE.  

proc compare base=derive.adsl compare=verify.v_adsl; 
run; 

 

%CompareWarn(); 

If the compare is clean, nothing happens. But if the compare has any issues, you will see log messages like 
the following: 

WARNING: %CompareWarn detects: BASE data set has observation not in COMPARE 

Or maybe: 

WARNING: %CompareWarn detects: A value comparison was unequal 

The main benefit in using the %CompareWarn macro is that it leverages the capabilities of modern SAS 
editors such as Enterprise Guide and Studio. Their color-coded Log Summary reports let you know that 
something is amiss in your compare even before you check the output.   

http://www.xianhuazeng.com/en/2016/11/26/removing-duplicate-texts-in-a-string/
https://github.com/srosanba/sas-comparewarn

If you work in an environment where log checking programs are used to make sure there are no ERROR or 
WARNING messages in your logs, this technique will also help to make sure there are no missed issues with 
broken compares as part of that log checking process. 

The %CompareWarn macro is available for download on GitHub: 

https://github.com/srosanba/sas-comparewarn 

This week’s tip was contributed by Shane Rosanbalm. Shane is a Senior Biostatistician at Rho, Inc., with a 
particular interest in SAS macro efficiencies, as well as graphical presentations of data.  

 

Weekly Tip for Dec. 22, 2020 

Get Fuzzy! Comparing Variables with Character and Numeric SAS Functions   

SAS® users often need to compare values of variables and assess the matches.  SAS has provided a number of 
tools which can perform “fuzzy matching”. Among these tools are “wild card” searches in a where statement 
using the LIKE (alias ?) and CONTAINS operators; string searches in an if statement using operators and 
functions; “fuzzy” comparison functions such as COMPGED and SPEDIS; PROC FCMP; PROC GEOCODE, and 
data-driven control tables enabling user-defined formats to standardize data. Our Tuesday Tip today 
introduces, by example, “fuzzy matching” tools and techniques, with an example of both character and 
numeric fuzzies. Additional information may be obtained in our paper, “Quick, Call the “FUZZ”: Using Fuzzy 
Logic”, at https://www.pharmasug.org/proceedings/2020/AP/PharmaSUG-2020-AP-007.pdf.  

Character strings, especially strings describing names and addresses, are notoriously dirty and prone to 
spacing, length and punctuation issues. Any real-world comparison of character strings or selection based on 
character strings needs to be both flexible and configurable, i.e., the degree of “sameness” needs to be 
quantifiable. SAS provides several character functions that allow you to make a fuzzy comparison: COMPARE, 
COMPGED, COMPLEV, SOUNDEX and SPEDIS. Each of these functions use a different fuzzy algorithm and can 
be used in conjunction with one another to achieve a (subjectively) optimal match. Use of these functions 
produces inexact results by definition, and results must be reviewed carefully.  

We illustrate an example of character fuzzies below, using the COMPGED function supplied by SAS. The 
COMPGED function determines how close the two arguments are in regard to matching (i.e., it determines 
the generalized edit distance between the two values). 

Syntax: COMPGED(string-1, string-2 <, cutoff > <, modifier(s)>) 

The cutoff is a numeric value that is returned if the generalized edit distance is greater than the cutoff value. 
The modifiers that can be used are the same as those used for COMPARE, described in the following table. 

https://github.com/srosanba/sas-comparewarn
https://www.pharmasug.org/proceedings/2020/AP/PharmaSUG-2020-AP-007.pdf

More information on the generalized edit distance can be found in the SAS documentation Functions and 
CALL Routines.  

Below we demonstrate how the generalized edit distance can be calculated based on differences in the two 
strings. 

data roster_compged; 
set newroster; 
FNCOMP = compged(FIRSTNAME, FIRSTNAME2); 
LNCOMP = compged(LASTNAME, LASTNAME2); 
FNCOMP_I = compged(FIRSTNAME, FIRSTNAME2, 'i'); 
LNCOMP_I = compged(LASTNAME, LASTNAME2, 'i'); 

run; 
 

FIRSTNAME LASTNAME FIRSTNAME2 LASTNAME2 FNCOMP FNCOMP_I LNCOMP LNCOMP_I 
Jan Write   Jan Wright 20 20 210 210 
Lucy Smyth Lucy Smith 0 0 100 100 
Kris Johnson Chris Johnson 300 300 0 0 
Chris Jones Chris Jones 0 0 0 0 
Tyler ones Tyler Jones 0 0 200 200 

Illustration of COMPGED function   

We see that for the first name the generalized edit distance without modifiers and with the modifier to 
ignore casing both yield a value of 20. This is because for FIRSTNAME2 there are two blank spaces that 
precede the value ‘Jan’. The unit cost for each blank space is 10 units, therefore, the total unit cost is 20 
units. For the last name we have to replace ‘gh’ with ‘te’ and truncate the value. The unit cost to replace a 
character is 100 units per character and the unit cost to truncate the output string is 10 units per character 
truncated. Since we had to replace two characters and truncate one character, the total unit cost is 210. 
Limits or cutoffs can be implemented as well. 

The technique demonstrated above, one of many designed for use with character strings, is highly useful, but 
what if we need to compare a numeric value? There are various options available for determining if a 
numeric value is equivalent to another numeric value. In some cases, the values will be exactly equal, and no 
additional comparison is needed. However, there are some cases where the values are not quite equal but 
are equal ‘enough’ so that if the values were rounded or truncated or a ‘fuzz’ factor is added then the values 
would be considered equal. Depending on the type of ‘fuzz’ factor you wish to consider will determine which 
function should be best utilized. Enter the FUZZ function! The FUZZ function either returns the nearest 

Modifier Description 

i or I Indicates that casing should be ignored for both strings. 

l or L Indicates that leading blanks should be removed from both strings before comparison. 

n or N Indicates that quotation marks from either argument should be removed and that casing 
should be ignored. 

: (colon) Indicates that the longer string should be truncated to the length of the shorter string. If th  
is not specified, then the shorter string is padded with blank spaces to the length of the 
longer string. 

 

https://documentation.sas.com/?docsetId=lefunctionsref&docsetTarget=p1r4l9jwgatggtn1ko81fyjys4s7.ht%20m&docsetVersion=9.4&locale=en
https://documentation.sas.com/?docsetId=lefunctionsref&docsetTarget=p1r4l9jwgatggtn1ko81fyjys4s7.ht%20m&docsetVersion=9.4&locale=en

integer (i.e., rounds up or down based on normal rounding rules) if the return value is within 1E-12 of the 
argument. If the return value is not within 1E-12 of the argument, then the FUZZ function returns the 
argument. 

data fuzz_score;  
set roster; 
format S_FUZZ 20.14;  
S_FUZZ = fuzz(SCORE);  

run; 
 

FIRSTNAME LASTNAME GENDER SCORE S_FUZZ 
Jan Write F 1.00000000000012 1.00000000000000 
Lucy Smyth F 1.00000000000121 1.00000000000121 
Kris Johnson F 1.00324325660746 1.00324325660746 
Chris Jones M 0.00000000000121 0.00000000000121 
Tracey Smith F 0.00000000000012 0.00000000000000 

Illustration of FUZZ function 

For the cell highlighted in pink, note that the return value was within 1E-12 and therefore, the value was 
rounded accordingly. 

We hope we have demonstrated that SAS has provided a myriad of tools to utilize for “fuzzy” matching by 
showing just two of many techniques. Selection of records with where statements (conditions and special 
operators) and if statements (:_ operator and functions); standardizing of records using fuzzy matching 
techniques including user defined formats, functions, and PROC GEOCODE (address information); and PROC 
FCMP (a user-defined function to clean addresses) are all discussed in the full paper referenced above. We 
hope you’ve gained some appreciation for the “fuzz” and you’ll get “fuzzy” along with us! 

This week’s tip was contributed by Richann Watson and Louise Hadden.  

Richann is an independent consultant. She is a statistical programmer and CDISC consultant with DataRich 
Consulting.  She is an active member of the ADaM team and co-leads the ADaM oncology team.  

Louise Hadden is a Lead Programmer Analyst in the Health and Environment Division at Abt Associates Inc., 
specializing in analysis and reporting. 

Weekly Tip for Jan. 05, 2021 

Purrfectly Fabulous Feline Functions   

SAS has a wide variety of useful functions and call routines that are constantly being added to and enhanced. 
Some of my favorites are the fabulous feline functions (CAT, CATS, CATT, and CATX) and three new CALL 
routines (CALL CATS, CALL CATT, and CALL CATX) that replace clunky and verbose concatenation routines.  

 

LENGTH stcounty $ 5 citystzip $ 60; 
stcounty=TRIM(LEFT(PUT(state,Z2.)))||TRIM(LEFT(PUT(county,Z3.))); 

citystzip=TRIM(LEFT(city))||','||TRIM(LEFT(statecode))||','|| 
               TRIM(LEFT(zip)); 

The customary syntax seen above consists of two functions (TRIM and LEFT) paired with the concatenation 
operator ( || .) TRIM removes trailing blanks, while LEFT left aligns text values. LEFT does not remove leading 
blanks. It simply moves leading blanks to the end of the string, which is why you usually see LEFT used in 
conjunction with (and inside) the TRIM function for string concatenation. Over the years, SAS has enhanced 
the function (pardon the pun) of TRIM and LEFT. A newer function TRIMN also strips trailing blanks: the 
difference between TRIM and TRIMN is the end result in the case of the argument being missing. TRIM 
results in a single blank (length 1) for a missing argument while TRIMN results in a null (length 0.) This might 
come in handy, for example, when concatenating first, middle and last names, in which middle names are 
frequently missing. Another newer function, STRIP, is the equivalent of TRIMN(LEFT(varx)) but is obviously 
more convenient. STRIP removes both leading and trailing blanks. The CAT CALLS and functions build on the 
original and derivative functions to offer a complete array of concatenation options - with a lot less code. 

    

Given the data set shown above, we can use CAT functions (or CAT calls) to create strings combining two or 
more variables. 

The CAT Function 

data temp; 
  set dd.harrypotter; 
  fullname=cat(firstname, middlename, lastname); 
run; 

 Result of using the CAT function: 

fullname 

Harry     James     Potter 

Note the spaces between the three words. The CAT function operates just like the concatenation bars, and 
simply strings variables together regardless of trailing blanks. Don’t forget to use a length statement (I 
used  $ 40) - the default length of the resulting string when using many character functions is 200. On the 
other hand, be careful - if you don’t leave sufficient length for the resulting string, you will get a warning 
about your buffer and the string will either be truncated or blank depending on your system. 

The CATS Function 
 

data temp; 
  set dd.harrypotter; 
  fullname=cats(firstname, middlename, lastname); 
run; 

  
Result of using the CATS function: 
 

fullname 
HarryJamesPotter 

With the CATS function, ALL blanks are STRIPPED from the resulting string. 

The CATT Function 

data temp; 
  set dd.harrypotter; 
  fullname=catt(firstname, middlename, lastname); 
run; 

  
Result of using the CATT function: 
 

fullname 
HarryJamesPotter 

With the CATT function, TRAILING blanks are removed from the resulting string. 

The CATX Function 

data temp; 
  length fullname $ 80; 
  set dd.harrypotter; 
  fullname=catx(' ',firstname, middlename, lastname); 
run; 

  
 
 
 
Result of using the CATX function: 
 

fullname 

Harry James Potter 

With the CATX function, EXTRA characters are added to the resulting string. 

The CATQ Function 

data temp; 
  length fullname $ 120; 
  set dd.harrypotter; 
  fullname=catq('CT',    
  combined_name, school, house); 
run; 

  
Result of using the CATQ function: 
 

fullname 
"Potter, Harry",Hogwarts,Gryffindor 

With the powerful CATQ function, a number of modifiers are available to customize the resulting string, 
including surrounding portions with quotation marks. 

CALL CATS, CALL CATT and CALL CATX all behave more or less like their corresponding functions described 
above, except that the resulting character variable is included in the CALL.  Arguments can be numeric 
instead of character;  numerics are converted to character using BESTw. Format.  Results for CAT functions 
and CAT CALLs are always character, and should not be reserved SAS variables. Items to be concatenated can 
be character or numeric. If an input item is numeric, it is converted to BESTw. format automatically and 
leading/trailing blanks are omitted. If an input item is blank and a delimiter is specified, CAT functions and 
CAT CALLs remove the delimiter for the blank variable. 

It is purrfectly clear that the CAT functions and call routines are valuable additions to the SAS® programmer’s 
toolbox.  Try them out, and you’ll become a CAT lover too! Please see the full paper for more information 
and a number of examples at https://www.pharmasug.org/proceedings/2018/EP/PharmaSUG-2018-
EP13.pdf. A poster and presentation are available upon request. 

 

This week’s tip was contributed by Louise Hadden. Louise Hadden is a Lead Programmer Analyst in the Health 
and Environment Division at Abt Associates Inc., specializing in analysis and reporting. 

  

https://www.pharmasug.org/proceedings/2018/EP/PharmaSUG-2018-EP13.pdf
https://www.pharmasug.org/proceedings/2018/EP/PharmaSUG-2018-EP13.pdf

Weekly Tip for Jan. 26, 2021 

Like, Learn to Love SAS® Like 

How do I like SAS®? Let me count the ways.... There are numerous instances where LIKE or LIKE statements 
can be used in SAS - and all of them are useful. This tip will walk through such uses of LIKE as: searches and 
joins with that smooth LIKE operator (and the NOT LIKE operator); the SOUNDS LIKE operator; using the LIKE 
condition to perform pattern-matching and create variables in PROC SQL; and PROC SQL CREATE TABLE LIKE.  

Smooth Operators 

SAS operators are used to perform a number of functions: arithmetic calculations, comparing or selecting 
variable values, or logical operations. Operators are loosely grouped as “prefix” (for example a sign before a 
variable) or “infix” which generally perform an operation BETWEEN two variables. Arithmetic operations 
using SAS operators may include exponentiation (**), multiplication (*), and addition (+), among others. 
Comparison operators may include greater than (>, GT) and equals (=, EQ), among others. Logical, or 
Boolean, operators include such operands as || or !!, AND, and OR, and serve the purpose of grouping SAS 
operations. Some operations that are performed by SAS operators have been formalized in functions. A good 
example of this is the concatenation operators (|| and !!) and the more powerful CAT functions which 
perform similar, but not identical, operations. Like operators are most frequently utilized in the data step and 
PROC SQL via a data step. There is a category of SAS operators that act as comparison operators under 
special circumstances, generally in where statements in PROC SQL and the data step (and DS2) and 
subsetting if statements in the data step. These operators include the LIKE operator and the SOUNDS like 
operator, as well as the CONTAINS and the SAME-AND operators.  

Character operators are frequently used for “pattern matching”, that is, evaluating whether a variable value 
equals / does not equal / sounds like a specified value or pattern. The LIKE operator is a case sensitive 
character operator that employs two special “wildcard” characters to specify a pattern: the percent sign (%) 
indicates any number of characters in a pattern, while the underscore (_) indicates the presence of a single 
character per underscore in a pattern. The LIKE operator is akin to the GREP utility available on Unix / Linux 
systems in terms of its ability to search strings.  

 

The LIKE operator, described above, searches the actual spelling of operands to make a comparison. The 
SOUNDS LIKE operator uses phonetic values to determine whether character strings match a given pattern. 
As with the LIKE operator, the SOUNDS LIKE operator is useful for when there are misspellings and similar 
sounding names in strings to be compared. The SOUNDS LIKE operator is denoted with a short cut ‘-*’. 
SOUNDS LIKE is based on SAS’s SOUNDEX algorithm. Strings are encoded by retaining the original first 
column, stripping all letters that are or act as vowels (A, E, H, I, O, U, W, Y), and then assigning numbers to 

groups: 1 includes B, F, P, and V; 2 includes C, G, J, K, Q, S, X, Z; 3 includes D and T; 4 includes L; 5 includes M 
and N; and 6 includes R. “Tristn” therefore becomes T6235, as does Tristan, Tristen, Tristian, and Tristin. 

 

Joins with the LIKE operator 

It is possible to select records with the LIKE operator in PROC SQL with a where statement, including with 
joins. For example, the code below selects records from the SASHELP.ZIPCODE file that are in the state of 
Massachusetts and are for a city that begins with “SPR”. 
  
proc sql; 
   create table temp1 as 
   select a.city, 
          a.countynm,  
          a.city2, 
          a.statename, 
          a.statename2  
   from sashelp.zipcode as a 
   where upcase(a.city) like 'SPR%' and pcase(a.statename)='MASSACHUSETTS'; 
quit; 
  
The test print of table TEMP1 shows only cases for Springfield, Massachusetts. 
 

CITY COUNTYNM CITY2 STATENAME STATENAME2 

Springfield Hampden SPRINGFIELD Massachusetts MASSACHUSETTS 

Springfield Hampden SPRINGFIELD Massachusetts MASSACHUSETTS 

Springfield Hampden SPRINGFIELD Massachusetts MASSACHUSETTS 

The code below joins SASHELP.ZIPCODE and a copy of the same file with a renamed key column (city  
geocity), again selecting records for the join that are in the state of Massachusetts and are for a city that 
begins with “SPR”. 

proc sql; 
   create table temp2 as 
   select a.City, 
          b.geocity, 
          a.countynm, 
          a.statename, 
          b.statecode, 
          a.x, 
          a.y   
   from sashelp.zipcode as a, zipcode2 as b 
   where a.city = b.geocity and upcase(a.city) like 'SPR%' and  
         b.statecode = 'MA'; 
quit; 
  

The test print of table TEMP2 shows only cases for Springfield, Massachusetts with additional variables from 
the joined file. 

CITY geocity COUNTYNM STATENAME STATECODE X Y 

Springfield Springfield Hampden Massachusetts MA -72.589584 42.099922 

Springfield Springfield Hampden Massachusetts MA -72.589584 42.099922 

Springfield Springfield Hampden Massachusetts MA -72.538580 42.116054 

The LIKE condition 

The LIKE operator is sometimes referred to as a “condition”, generally in reference to character comparisons 
where the prefix of a string is specified in a search. LIKE “conditions” are restricted to the data step, as the 
colon modifier is not supported in PROC SQL. The syntax for the LIKE “condition” is: 
  
  where firstname=: 'Tr'; 

This statement would select all first names in Table 1 above. To accomplish the same goal in PROC SQL, the 
LIKE operator can be used with a trailing % in a where statement. 

PROC SQL CASE LIKE 

The LIKE operator in PROC SQL can be used in conjunction with the CASE WHEN statement to create 
variables. The code snippets below feature using the LIKE operator with OR operators and CASE WHEN to 
create both a categorical variable (DXTYPE) and binaries (CANCERDX, ARTHDX, BCKPNDX, RHEUMDX, 
NONMSTDX, and OVERDOSE) from selected diagnosis codes with varying prefix lengths. (Please note that the 
list of diagnosis codes for each category or binary is incomplete – this code snippet is for demonstration 
purposes only.) 

PROC SQL; 
 CREATE TABLE out1 AS   
 SELECT DYAD_OPIOID_ID, 
        analelig, 
        DIAG,  
        CASE 
        WHEN DIAG LIKE '14%' 
        OR diag  LIKE 'D45%' 
        THEN 'CANCERDX' 
            WHEN DIAG LIKE '720%' 
        OR diag  LIKE 'M6788%' 
        THEN 'BCKPNDX' 
        WHEN DIAG LIKE '725%' 
        OR diag  LIKE 'R2989%' THEN 'RHEUMDX' 
            WHEN DIAG LIKE '3382%' 
        OR diag  LIKE 'T50%' THEN 'OVERDOSE' 
        ELSE '' 
        END AS DXTYPE, 
           CASE 
           WHEN DIAG LIKE '3382%' 
           OR diag like 'G89%' 
           THEN 1 
           ELSE 0 
         END AS NONMSTDX, . . . 

PROC SQL CREATE TABLE LIKE 

There are several methods of creating an empty data set. PROC SQL CREATE TABLE LIKE is one of the most 
efficient ones when creating a shell from an existing data set, as it automatically uses the metadata from the 
existing SAS data set. The CREATE TABLE LIKE method is NOT benign and will overwrite an existing data set. 
Therefore, the program uses conditional logic and %SYSFUNC to determine if the file to be written already 
exists, and does not overwrite the file, instead producing an error note in the log.  

%if %sysfunc(exist(annual.allratings&fileyear._long))=0 %then %do; 
  PROC SQL; 
 CREATE TABLE annual.allratings&fileyear._long 
         LIKE prevann.allratings&prevyear._long; 
 DESCRIBE TABLE annual.allratings&fileyear._long; 
  QUIT; 
%end; 
  
%else %do;  
   %put ERROR: Data Set ALLRATINGS&FILEYEAR._LONG already exists!; 
 run; 
%end; 

SAS provides practitioners with several useful techniques using LIKE statements: the smooth LIKE OPERATOR 
/ CONDITION in both the data step and PROC SQL, CASE WHEN LIKE in PROC SQL, and CREATE TABLE LIKE in 
PROC SQL. There are definitely reasons to like LIKE in SAS programming.  

This week’s tip was contributed by Louise Hadden. Louise Hadden is a Lead Programmer Analyst in the Health 
and Environment Division at Abt Associates Inc., specializing in analysis and reporting.  

Weekly Tip for Feb. 2, 2021 

SAS® formats are a ubiquitous part of every SAS programmer’s toolbox. Read on for some useful tips on 
how to deal with embedded formats that aren’t included in a data submission, and how to gracefully 
ensure that your data deliveries do not cause similar problems for users! 

SAS® practitioners are frequently called up to format variables in SAS data sets they have received or 
created for various use cases. Analysts and other end users desire the convenient categorization, 
transformative nature, and attractive appearance that SAS formats can lend to variables for reports and 
further analytic and data set construction purposes. SAS formats can be created in SAS workspace, and can 
be stored permanently in SAS catalogs, a specially purposed container for SAS files. SAS formats created in 
workspace are ephemeral, and only exist for the duration of a SAS session. SAS formats stored in a SAS 
catalog are notoriously difficult to transfer across platforms, SAS versions and “bit” versions (32-bit vs 64-
bit). Recipients of SAS data sets with “embedded formats” and/or SAS catalogs originating from 
incompatible systems find themselves in a quandary – SAS reports errors when it can’t find a compatible 
catalog (IF a catalog accompanies a data set) for user defined formats permanently associated with variables 
in a SAS data set. SAS catalogs are also very difficult to update, document and manipulate. We propose 
some straightforward SAS solutions for the creation, transfer and use of SAS formats. 

The bane of every SAS programmer’s existence are mystery data sets: those data sets with unknown ancestry 
and little to no information regarding the contents of the data set, delivered without the comfort of well-
written documentation and background information such as a survey instrument. Most of us, when faced 
with such a “mystery box”, resort to PROC CONTENTS, a test print, maybe some freqs and means. If you are 
lucky, you gain useful information from this strategy. If you are REALLY lucky, you can proceed with your 
analytic and/or file building tasks. All too often, it’s a worst-case scenario: you may not have noticed the 
presence of user-defined formats in the PROC CONTENTS output in your testing, and/or you went straight to 
a test print or frequencies. The next thing you know you see the dreaded error message in your log: 

 

ERROR: The format $YNDKF. was not found or could not be loaded. 

Reality comes crashing in. Sometimes SAS data sets are delivered with embedded formats, and the expected 
format catalog is either not supplied, is incompatible due to system/platform/bit differences, is outdated, is 
corrupted, and/or is incomplete. 

OPTIONS nofmterr; 

This little SAS system option is a quick band aid that will allow you to open and process a data set with 
embedded formats that you do not have a catalog for, but you will not have the benefit of the formats or 
view the data as it was intended to be. 

OPTIONS FMTSEARCH=(lib.survey1 lib.survey2 lib2.survey3); 

 

At other times, the problem may be the location of the missing library. Using the fmtsearch option allows you 
to load format catalogs in the order that they are mentioned. Note that the librefs are tied to the libname 
statements in your program. 

But what do you do if you STILL don’t have a format catalog for your data set? 

If the data set that has been delivered without being properly attired is from a known source that has been 
contracted to provide the data, request that incoming data sets be delivered without embedded formats and 
with well documented programs to (a) create a collection of formats and (b) associate those formats with 
variables in the delivered data set(s). 

 

If you are unable to obtain better documentation and/or your format catalog, not all is 
lost. Strip embedded formats with the option NOFMTERR described above and a data 
step by simply reformatting variables with an “empty” format statement.  

DATA WANT; 
   SET HAVE; 
   FORMAT _ALL_; 
RUN; 

 

PROC DATASETS, ATTRIB statements or PROC SQL also work well to (a) create a collection of formats and (b) 
associate those formats with variables in the delivered data set(s). Documentation of the data set(s) 
certainly helps in this effort. 

Strip embedded formats and (re) build formats and format associations using techniques described in detail 
below. 

 

Now we have associated formats in our file - what else should we do? 

Create transportable, modifiable formats to associate with variables in data sets of in 
procedures. 

Build a SAS data set to house elements of a format catalog that can be transformed into catalogs via PROC 
FORMAT CNTLIN. 

PROC FORMAT LIBRARY = LIBRARY.CHILD_DAY CNTLIN=PRVTYPID; RUN; 

Build a SAS data set from an existing SAS format catalog via PROC FORMAT CNTLOUT that can be used for 
PROC FORMAT CNTLIN.  

Just as you can use PROC FORMAT CNTLIN to read a SAS data set into a temporary or permanent format 
catalog, you can also get information OUT of a format catalog. When you (or someone else) create a format 

catalog, even in workspace, a SAS data set is generated with an inordinate number of columns, of which 
only five or six are commonly used, FMTNAME, TYPE, START, END, LABEL and HLO (only three are required – 
FMTNAME, START and LABEL). 

PROC FORMAT = LIBRARY.CHILD_DAY;  
   CNTLOUT = CNTLOUT; 

    RUN;  
    QUIT; 

WYO – write your own format! 

PROC FORMAT = LIBRARY.CHILD_DAY;  
VALUE YESNODK 1 = "Yes" 

2 = "No" 
8 = "Don't Know"; 

 RUN; 
 

Additionally, you can use an include statement that contains PROC FORMAT statements to read in formats 
within your SAS program, and then apply them to a work catalog so that they can be used. If you can locate 
an electronic survey instrument or existing documentation, it’s possible to process these files to create a 
format creation program. 

When building SAS format catalogs, use identifiable two-part format catalog names i.e., survey1.sas7bcat, 
survey2.sas7bcat instead of formats.sas7bcat. To save a format or formats in a permanent format catalog, 
use PROC FORMAT to create a format by either defining value statements or by using PROC FORMAT’s 
CNTLIN statement. Note the use of a specific format catalog name instead of the default formats.sas7bcat, 
which makes it easier to identify the specific formats needed for a task. In addition, if you are creating several 
different types of catalog entries for a project, make sure to distinguish between format, macro, and 
template catalogs either by name or by locating them in separate folders to avoid overwriting your entries. 

The below sample shows how to modify a format by adding a description, and the PROC CATALOG listing 
shows the entry name, type, dates and description. 

PROC CATALOG CATALOG = LIBRARY.CHILD_DAY; 
   MODIFY PRVTYPID FORMAT (DESCRIPTION = "Categories for Provider Type ID"); 
   CONTENTS; 
RUN; 
QUIT; 

The PROC CATALOG CONTENTS statement lists the contents of the format catalog. Note that PROC CATALOG 
is the ONLY way to add a description to a format, as PROC FORMAT does not have a LABEL or DESCRIPTION 
option.  

 

For more detail on FORMAT type entries, use PROC FMTLIB. Note that PROC CATALOG provides a different set 
of information from PROC FORMAT FMTLIB, and that both outputs are useful. 

PROC FORMAT CATALOG = LIBRARY.CHILD_DAY FMTLIB; 
RUN; 

Contents of Catalog LIBRARY.CHILD_DAY 
# Name    Type   Create Date  Modified Date  Description 
1 PRVTYPID  FORMAT 07/21/2019   07/21/2019   Categories for Provider Type ID 

A very powerful technique is to build a SAS data set instead of a SAS format catalog to house your formats. 

* 1: map provnum to fips county code; 

 
data ctrl_prov_fips; 
length label $ 3 BASEVAR $ 32; 
    set prov_xwalk2 (keep = provnum county_fips 
    rename=(provnum=start county_fips=label)) end=last; 
    retain fmtname '$FIPSCTY' type 'c' BASEVAR ‘PROVNUM’; 
    output; 
    if last then do; 
    hlo='O'; 
    label='***'; 
    output; 
    end; 
    keep fmtname type start label hlo basevar;  
run; 
     
proc sort data = ctrl_prov_fips ;  
    by fmtname start; 
run; 
 

* Put the formats together, setting length long enough to accommodate 
the longest label; 

 
data allgeogformats_&fileyear.&filedate.; 
   length fmtname $8 type $1 start $14 label $255; 

    set ctrl_prov_fips 
        ctrl_prov_cntynm 
        ctrl_prov_cbsa 
        ctrl_prov_urban 
        ctrl_prov_regionc 
        ctrl_prov_fips_st; 
    by fmtname start; 
    label fmtname = 'Name of Format' 
          start = 'Start of Range for Format' 
          label = 'Label for Format' 
          type = 'Type of Format'; 
run; 

FORMAT NAME: PRVTYPID LENGTH: 24 
MIN LENGTH: 1 MAX LENGTH: 40 DEFAULT LENGTH: 24 FUZZ: STD 
START         |END         |LABEL (VER. 9.4 24MAR2019:13:35:29) 
---------------- +-------------  +------------------------------------ 
CCC |CCC |Child Care Center 
HS  |HS |Healthy Start 
AR        |AR |At Risk 

OSH      |OSH |Outside of School Hours 

You can then use this data set for reporting and/or to create a format statement, as 
shown below. 
proc format library = work  
    cntlin = allgeogformats_&fileyear.&filedate.; 
run; 

We hope you can use some of these tips to declare your independence from the format catalog! 

This week’s tip was contributed by Nancy McGarry. Nancy is a Lead Programmer Analyst in the Social and 
Economic Policy Division at Abt Associates Inc., specializing in nutritional analysis and reporting. 

Weekly Tip for Mar. 16, 2021 

Order, Order! Seven Ways To Reorder Your Variables, Ranked by Elegance and Efficiency 

SAS® practitioners are frequently required to present variables in an output data file in a particular order, or 
standards may require variables in a production data file to be in a particular order. Fortuitously, SAS 
provides us with numerous methods to control all types of SAS output, including SAS data files, data tables in 
other formats, and ODS output. In this Tuesday Tip I’ll focus solely on output SAS data files (which may then 
be used to generate SAS data files and other types of output from SAS processes), and specifically on DATA 
step and PROC SQL methods. 

It is always important to fully understand and explore the inputs to SAS-created output. Important concepts 
include considering the processes by which SAS assigns variable attributes, including variable/column order 
within a data file, and how to obtain information on variable attributes.  SAS has provided users with a 
variety of possibilities in terms of determining locations of variables or columns (and other important details 
comprising metadata). These possibilities include, but are not limited to: 

1.  The CONTENTS Procedure 

2.  The DATASETS Procedure 

3.  SAS Dictionary Tables (used with PROC SQL) 

4.  SASHELP Views (can be used with PROC SQL and with the SAS DATA step) 

5.  %SYSFUNC (through MACRO functions) 

6.  “V” functions (through data file functions) 

Any one of the processes listed above can produce the location of a given variable, which is our primary interest 
here. 

* THE CONTENTS PROCEDURE; 
PROC CONTENTS DATA=DD.HARRYPOTTER_LABELLED 
    OUT=LABELLED_CONTENTS 
 (KEEP=NAME TYPE LENGTH LABEL FORMAT INFORMAT VARNUM); 
RUN; 

For the purposes of expediency, an abridged CONTENTS procedure listing of a test data file is provided 
below, showing the column header “#”, which is the label for the metadata variable “VARNUM”. Note that an 

output data file is created from the CONTENTS procedure, with a KEEP statement that retains only the 
variables of interest.  

Variables in Creation Order 

# Variable Type Len Label 

1 combined_name Char 50 Full Name 

2 firstname Char 25 First Name 

7 patronus Char 25 Patronus 

8 animal Char 25 Pet 

9 school Char 50 School 

16 deatheater Num 8 Binary: Death Eater 

17 bloodstatus Char 10 Blood Status: Halfblood, Muggleborn, Squib, Pureblood 
 

It is important to have an understanding of how SAS handles and defines records and variables in the data 
file, i.e., through the Program Data Vector (PDV), so as to manipulate data file attributes. The PDV builds data 
file observations during processing, storing data values in memory, and finally writes out both an output data 
file and a metadata data file containing variable attributes when a data file compiles. 

In the listing above, 5 columns are displayed, including a column for the attribute of interest, VARNUM. The 
CONTENTS procedure only displays variable attributes that are present in the data file being analyzed, so a 
contents listing for a data file that does not have any assigned formats, informats or indices will not have 
columns for those attributes. 

If we wish to change the order in which variables are presented in an existing data file, we need to know how 
SAS determines variable order under various conditions. In the DATA step, variable order is determined by 
the first mention of the variable as it passes through the PDV. DATA step statements that can set the order of 
variables include ARRAY, ATTRIB, FORMAT, INFORMAT, LENGTH and RETAIN. These statements MUST 
precede any SET, MERGE or UPDATE statements so as to accomplish the goal of reordering variables. The 
order in which variables are mentioned in the statement determines the order of the variables in the 
corresponding data file. If variables are not listed in the statement, variables that are mentioned will appear 
in the order that they are mentioned, remaining variables will appear in their original order in the originating 
data file (or INPUT statements in the case of data files being created). 

In addition to DATA step statements, PROC SQL can also be utilized to reorder variables in an existing data 
file, in a similar fashion to the DATA STEP statements described above. PROC SQL SELECT and SELECT AS 
statements may be used to order, format, set the length of and rename variables in a new data file created 
from the existing data file. 

Common reasons for wanting to reorder variables in a data file are the result of merging files (order will be 
common variables first, then variables in the original order of each merged file) or the use of ARRAY, ATTRIB, 
LENGTH or RETAIN statements in originating data files that “disordered” variables. 

Building reordering data set statements programmatically using SAS metadata 

Examples of the data step and PROC SQL reordering techniques will be discussed below. Each method of 
reordering data requires the provision of metadata information in the statement. Rather than type in what 
could be a very extensive variable attribute list, SAS metadata can be used to build the statements. Some 
methods (RETAIN, FORMAT, INFORMAT, PROC SQL) require an ordered list of variables as input (note that 
additional metadata information can be used, but is not strictly necessary.) Others (ATTRIB, ARRAY, LENGTH) 
require additional information, such as the length of a variable, variable label, and/or type of variable. SAS 
metadata can be used to construct the statements in a variety of ways. Two methods of harnessing SAS 
metadata to build data-driven code follow. The precursor to these steps is the use of the 
LABELLED_CONTENTS data file created above, and sorted in various ways into different output data files. 

* TEST FILE PRECURSOR STEPS: SORT OUT DATA SETS IN SPECIFIC ORDERS; 
* VARIABLE NAME (NAME) IS KEY; 
* POINT IS TO PROVIDE VARIABLE LISTS IN DIFFERENT ORDER; 
  
*ALPHA ORDER; 
  
PROC SORT DATA=LABELLED_CONTENTS OUT=LABELLED_BY_NAME; 
 BY NAME; 
RUN; 
  
*DESCENDING POSITION; 
  
PROC SORT DATA=LABELLED_CONTENTS OUT=LABELLED_BY_DVARNUM; 
 BY DESCENDING VARNUM; 
RUN; 
  
*LENGTH ORDER; 
  
PROC SORT DATA=LABELLED_CONTENTS OUT=LABELLED_BY_LENGTH; 
 BY LENGTH; 
RUN;  
  
* TEST FILE PREPARATION METHOD 1: MACRO LISTS FROM METADATA VIA PROC SQL INTO:; 
* CREATES MACRO LISTS THAT CAN BE USED IN DATA SET STATEMENTS AND PROC SQL; 
  
PROC SQL; 
  SELECT NAME 
  INTO :VARNUM_ORDER SEPARATED BY ' ' 
  FROM LABELLED_CONTENTS; 
quit; 
  
%PUT By Original Order = &VARNUM_ORDER; 
  
PROC SQL; 
  SELECT NAME 
  INTO :NAME_ORDER SEPARATED BY ' ' 
  FROM LABELLED_BY_NAME; 
quit; 
  
%PUT By Alpha Order = &NAME_ORDER; 
  
* TEST FILE PREPARATION METHOD 2: STRUCTURED INCLUDE FILE VIA DATA _NULL_; 
* CREATES AN “INCLUDE” FILE TO BE USED IN LENGTH STATEMENTS USING SAS METADATA; 
* NOTE CONDITIONALLY CREATES STRINGS BASED ON TYPE; 
  
DATA _NULL_; 
  FILE '.\STRING_ORDER.TXT' LRECL=50 PAD; 
  LENGTH STRING_ORDER $ 50; 
  SET LABELLED_BY_NAME; 
  IF TYPE=2 THEN STRING_ORDER=CATX(' ',NAME,'$',LENGTH); 

  ELSE IF TYPE=1 THEN STRING_ORDER=CATX(' ',NAME,LENGTH); 
  PUT STRING_ORDER; 
RUN; 
  
* TEST FILE PREPARATION METHOD 3: STRUCTURED INCLUDE FILE VIA DATA _NULL_; 
* CREATES AN “INCLUDE” FILE TO BE USED IN ARRAY STATEMENTS USING SAS METADATA; 
* NOTE CONDITIONALLY CREATES SEPARATE MACRO VARIABLES BASED ON TYPE; 
  
DATA CHARS NUMS; 
 SET LABELLED_BY_DVARNUM; 
 IF TYPE=2 THEN OUTPUT CHARS; 
 ELSE IF TYPE=1 THEN OUTPUT NUMS; 
RUN; 
  
PROC SQL; 
  SELECT NAME 
  INTO :DVARNUM_ORDER_C SEPARATED BY ' ' 
  FROM CHARS; 
QUIT; 
  
PROC SQL; 
  SELECT NAME 
  INTO :DVARNUM_ORDER_N SEPARATED BY ' ' 
  FROM NUMS; 
quit; 
  
%PUT By Descending Varnum Char Order = &DVARNUM_ORDER_C; 
%PUT By Descending Varnum Num Order = &DVARNUM_ORDER_N; 

LENGTH and ATTRIB statements 

An example of using the LENGTH statement (created as a %INCLUDE file in test file preparation method 2) is 
shown below. The LENGTH statement requires that a type designation (for character variables) and length be 
provided, thus SAS metadata was used in the creation of the %INCLUDE file above. Note that the LENGTH 
statement is above the set statement. The ATTRIB statement can be constructed in a similar fashion as the 
LENGTH statement using SAS metadata, and employed in the same manner. 

* EXAMPLE 1: LENGTH STATEMENT (ATTRIB STATEMENT IS SIMILAR); 
  
DATA DD.HARRYPOTTER_LENGTH 
 (LABEL="Harry Potter Sample Data Set - Variables Ordered with Length 
  Statement"); 
  LENGTH 
  %INCLUDE '.\STRING_ORDER.TXT'; 
  SET DD.HARRYPOTTER_LABELLED; 
RUN; 
  
*PROC CONTENTS ON OUTPUT DATA SET; 
  
PROC CONTENTS DATA=DD.HARRYPOTTER_LENGTH VARNUM; 
RUN; 

 

ARRAY statements 

Using ARRAY statements created by using SAS metadata is similar to the LENGTH statement. Separate macro 
variable lists of character and numeric variables are created based on variable type in the metadata and 
deployed in the DATA step, again above the SET statement. The reason for separating character and numeric 

variables is because arrays do not allow mixing of variable types. Additional instructions can be put on the 
ARRAY statements such as format information, and depending on the metadata values for type and format, 
you could have multiple ARRAY statements. In this example, no information beyond type is utilized. 

* EXAMPLE 2: ARRAY STATEMENT; 
  
DATA DD.HARRYPOTTER_ARRAY 
 (LABEL="Harry Potter Sample Data Set - Variables Ordered with Array 
 Statements"); 
 ARRAY CH (*) $ 50 &DVARNUM_ORDER_C. ; 
 ARRAY NU (*) &DVARNUM_ORDER_N. ; 
 SET DD.HARRYPOTTER_LABELLED; 
RUN; 
  
PROC CONTENTS DATA=DD.HARRYPOTTER_ARRAY VARNUM; 
RUN; 
  

 RETAIN / FORMAT / INFORMAT statements 

The RETAIN statement is similar to the ARRAY statement in that variable metadata information is not 
required to reorder variables, just a list of the variables in the appropriate order. The FORMAT and 
INFORMAT statements behave in the same fashion. All three of these DATA STEP statements can perform 
additional operations, but in this context, they are not necessary and can complicate the task. For example, 
RETAIN is used to retain, or carry values across iterations of a DATA step. If RETAIN is used for this purpose, 
AND to reorder the data, there’s a risk of errors and/or unexpected results. The ability of these statements to 
reorder  variables is simply a by-product of how they operate. 

* EXAMPLE 3: RETAIN / FORMAT / INFORMAT STATEMENTS; 

  
DATA DD.HARRYPOTTER_RETAIN 
 (LABEL="Harry Potter Sample Data Set - Variables Ordered with Retain 
 Statement"); 
 RETAIN &NAME_ORDER.; 
 SET DD.HARRYPOTTER_LABELLED; 
RUN; 
  
PROC CONTENTS DATA=DD.HARRYPOTTER_RETAIN VARNUM; 
RUN; 

 PROC SQL 

PROC SQL in the context of variable reordering performs in a similar fashion to the RETAIN / FORMAT / 
INFORMAT statements in that PROC SQL only REQUIRES a reordered list of variables, but allows the 
specification of additional attributes, such as type, length, format, etc. PROC SQL is also unique in that it 
allows you to rename or “alias” variables. A common reason for reordering variables in a data file to make 
reporting more straightforward. However, there are other circumstances that require that output data 
(temporary or permanent SAS data files, Excel files, XML files, etc.) be ordered in a certain way. For example 
for XML files that are created using XML maps or schemas, case (upper, lower, proper or camelcase), matters 
as well as variable order. PROC SQL allows for the renaming of variables to match XML case requirements in 
the reordering process. 

Although the code for this example was not built programmatically, as with the other three examples, it 
could easily be. 

* EXAMPLE 4: PROC SQL TO REORDER VARIABLES; 
  
PROC SQL; 
 CREATE TABLE DD.HARRYPOTTER_SQL AS SELECT 
 animal                                     
 ,bloodstatus                                
 ,combined_name                              
 ,da                                            
 ,deatheater                                    
 ,deceased as deadasadoornail                                        
 ,firstname                                   
 ,house                                      
 ,lastname                                    
 ,middlename                                  
 ,ministryofmagic as MOM                              
 ,nickname                                    
 ,oop as OrderOfthePhoenix                                          
 ,patronus                                    
 ,professor                                      
 ,quidditch_wins                                 
 ,school   
 from dd.harrypotter_labelled 
 ORDER BY lastname ; 
QUIT; 

As you can see, SAS metadata coupled with an understanding of how the Program Data Vector (PDV) works 
allows programmers to construct both documentation of data set and variable attributes, and data-driven 
code to modify data files. These modifications include reordering variable/column order within an output 
data file. Methods to reorder variables described are the use of six specific data file statements and PROC 
SQL. The PROC SQL and the ATTRIB statement methods provide the most flexibility and least amount of time 
spent coding to accomplish the functional outcome of reordering variables. PROC SQL edges out the ATTRIB 
statement with the ability to rename or “alias” variables within the same procedural statement. 

This week’s tip was contributed by Louise Hadden. Louise Hadden is a Lead Programmer Analyst in the Health 
and Environment Division at Abt Associates Inc., specializing in analysis and reporting.  

 

Weekly Tip for Mar. 23, 2021 

Tips and Tricks with the SAS Macro Facility: Deleting Macro Variables 

In SAS DATA STEP programming, users can manage SAS variables on a data set through the DROP and KEEP 
statements, or respective data set options. What about macro variables? Does this capability extend to the 
SAS macro facility? 

With %PUT, programmers can check and validate macro variables and values, and print macro variables with 
their current values in the SAS log. In a %PUT statement, you can specify single macro variables, multiple 
macro variables, or categories of macro variables, such as automatic, user-defined, global and local. The 
example below uses %PUT _ALL_ to print all existing macro variables to the log; 

 %PUT _ALL_; 
     GLOBAL CLIENTMACHINE 10.0.2.2 
     GLOBAL GRAPHINIT   
     GLOBAL GRAPHTERM   
       .      
       .        
       .  
     AUTOMATIC SYSTIME 19:04 
     AUTOMATIC SYSUSERID sasdemo 
     AUTOMATIC SYSVER 9.4    

The SAS Macro facility features capabilities to help you manage macro variables during a SAS session. In the 
SAS community, much attention is given to the methods used to create and define macro variables; %LET, 
CALL SYMPUT, and PROC SQL INTO:. With %SYMDEL, macro variables can be deleted during a SAS session. 
Deleting macro variables will be valuable in code that contains nested macro calls, and when executed will 
generate multiple local symbol tables. Also, deleting macro variables may be necessary to prevent conflicts 
between global and local macro variables which share the same name. 

%SYMDEL can be used in open-code, or inside a macro definition. It will only delete global macro variables, 
however. %SYMDEL can be used to delete single macro variables, as well as multiple macro variables. In 
contrast to %PUT,  %SYMDEL cannot delete classes of macro variables through the use of keywords (_ALL_, 
_GLOBAL_, etc.).  

This Tuesday Tip provides a technical solution to the issue of deleting groups of macro variables, using the 
SASHELP.VMACRO view. The SASHELP view VMACRO contains metadata on macro variables. As shown 
below, the view contains a list of defined and available macro variables in a current SAS session.  

Scope       Name   Offset  Value 
GLOBAL       CLIENTMACHINE 0  10.0.2.2 
GLOBAL       GRAPHINIT  0  
GLOBAL       GRAPHTERM  0  
GLOBAL       SASWORKLOCATION 0 
"/tmp/SAS_workC27800002E3E_localhost.localdomain GLOBAL SYSCASINIT 0  0 
GLOBAL       SYSSTREAMINGLOG 0  True 
GLOBAL       SYSUSERNAME  0  Sasdemo 
GLOBAL       SYS_SQL_IP_ALL 0  -1 
GLOBAL       SYS_SQL_IP_STMT 0  

Using the PROC SQL INTO clause with a SEPARATED BY option, developers can store a concatenated list of 
macro variable names in a macro variable. The example below demonstrates this task using the respective 
variables from the VMACRO view in the SASHELP library and inserting it into a macro variable named 
MACVNAME. 

PROC SQL NOPRINT; 
    SELECT Name INTO : MACVNAME SEPARATED BY  ' ' 
    FROM SASHELP.VMACRO 
    WHERE Scope='GLOBAL' AND  
     Name NOT IN ('SYSCASINIT', 'SYSSTREAMINGLOG', 'SYSUSERNAME',  
         'SYS_SQL_IP_STMT', 'SYS_SQL_IP_ALL' 'GRAPHTERM'); 
QUIT; 

%SYMDEL &MACVNAME MACVNAME; 
 
%PUT _GLOBAL_; 

In the code, I exclude six macro variables from the macro variable list which are necessary to run a SAS 
session. These macro variables cannot be deleted in SAS Studio or SAS University Edition. Using the scope 
variable, I subset the query to global macro variables exclusively. I place the call to the MACVNAME macro 
variable in the %SYMDEL statement. I also delete the MACVNAME macro variable to prevent the list from 
cluttering up the log.  

 85         %SYMDEL &MACVNAME MACVNAME; 
 86          
 87         %PUT _GLOBAL_; 
  
 GLOBAL GRAPHTERM   
 GLOBAL SYSCASINIT 0 
 GLOBAL SYSSTREAMINGLOG true 
 GLOBAL SYSUSERNAME sasdemo                                                                                                                                                                                          
 GLOBAL SYS_SQL_IP_ALL -1 
 GLOBAL SYS_SQL_IP_STMT  

The excerpt from the SAS log confirms the macro variables have been deleted from the global symbol table. 

This week’s tip was contributed by Jay Iyengar. Jay Iyengar is a SAS Certified Advanced Programmer, and SAS 
consultant with Data Systems Consultants LLC. 

Weekly Tip for May 4, 2021 

Generation of Clinical Reports Utilizing SAS Programming with PROC REPORT 

Patient profiles are used to display the clinical study data that is collected for an individual patient, organized 
by time and data category. Patient profiles are often used to report and populate clinical study information 
provided to the FDA as part of a regulatory submission. 

Automation is often the goal of SAS programming. The SAS Macro facility can expedite the process for the 
same. We can equip our macros with logic, we can feed them different parameters and run as they execute 
one output after the other. 

The process discussed in this Tuesday Tip  is a development methodology employed in the generation of the 
patient profile requirements and patient profile application architecture. It further demonstrates that SAS 
programmers can easily generate and automate diverse clinical report generation. The method discussed 
further focuses on the patient profile generation utilizing SQL Procedures and Macro Generation. 

Description: 

Our macro call looks something like: 

%macro clinreports(); 

proc sql noprint; 
  select count(*) into :subjCount from work.Headerrecord; 
quit; 
  
data _null_; 
  set work.Headerrecord(firstobs=&iter. obs=&iter.); 
  call symputx("usubjid",'USUBJID'n) ; 
  call symputx("SITENAME",'SITENAME'n);  
run; 
  
/***Setup output path;**********/ 
 
ods listing close; 
ods escapechar '^'; 
options orientation=landscape nodate nonumber nobyline nocenter; 
ods noresults; 
 
ods tagsets.rtf file ="&outpath\Patient Profile 16_Dec\&usubjid._&datacut..rtf"  
                style=minimal uniform; 
 
/***********Basic Header & Footer Information*************/ 
 
title1 height=1 j=l "&company." j=r "Site Name : %sysfunc(strip(&SITENAME.))"; 
footnote1 height=1 j=l "Dataset Date: &CRDATE."  
        j=r "Created on: %sysfunc(date(),date11.) %sysfunc(time(),hhmm.) (Confidential)"; 
 
/***Domain : Demographics***********/ 
  
proc report data=DMDIS center nowd headline headskip spacing=1 split="|"  
     missing style(header)={background=pink} style(report)={outputwidth=100%}; 
  column (' ^R/RTF"\i\b\fs18\cf2 Demographics' page ACTARM AGE); 
  define page   / order noprint; 
  define ACTARM / display  style(header)={just=left vjust=top}     
                  style(column)={cellwidth=7% just=left vjust=bottom}; 
  define AGE    / display  style(header)={just=left vjust=top}    
                  style(column)={cellwidth=3% just=left vjust=bottom};  
run; 
ods tagsets.rtf close; 
  
%let iter=%eval(&iter.+1); 
  
%end; 
  
%mend; 
  
%clinreports; 

This flexible report generation technique will allow the %clinreports macro to write PROC REPORT code 
dynamically. One can write the PROC REPORT using this macro and also add the variables needed per domain 
in the DEFINE statements. The clinical patient profile is generated per subject id in the clinical study by calling 
the subjcount macro variable during the macro execution. In fact, there can be many DEFINE statements in 
our PROC REPORTs, and no hard coding or tedious macro logic is necessary. Moreover, adding new clinical 
proc reports with this approach becomes as easy as entering new parameters. You add the macro call, SAS 
does the rest. 

The diverse clinical report generation requires the data sets to be pre-programmed, sorted and ready for 
reporting by the macro calls are executed. It is optimized for clinical patient profiles and clinical data listings. 
The SAS program provided above can be applied to a variety of situations and it can further improve the 
review process.  

This week’s tip was contributed by Ballari Sen. Ballari is a Clinical Data Programmer at Agios Pharmaceuticals, 
Inc., and specializes in SDTM Data Development, Clinical SAS Programming and Clinical Data Management. 

Weekly Tip for May 11, 2021 

Turn the Tables on Boring Reports with SAS and TAGSETS.RTF Options 

This Tuesday Tip highlights a technique to rotate column headers in tabular RTF output without 
postprocessing in Microsoft Word or Open Office. Prior to SAS 9.2, this could only be accomplished using a 
postprocessing macro within SAS. Post SAS 9.2, rotating column headers can easily be accomplished simply 
by specifying RTF tagset OPTIONS.  
 
RTF, like HTML, is an incredibly complex markup language which contains formatting commands. An 
RTF file can be viewed in a text editor as a text file. (See screenshot below for an example.) 
Microsoft has its own specification containing RTF commands, which is massive and overwhelming in 
scope. As you might guess, this specification changes for each version of Microsoft Word. You can download 
these specifications from the Microsoft download page. 
 

 
 

The PRINTER TAGSET / RTF destination combination was and is useful, but in order to take RTF output into a 
new dimension (i.e., out of the horizontal plane), SAS developed an RTF “measured” TAGSET. The new RTF 
tagset made a number of things possible that were impossible before and made other things (like rotation of 
column headers) a lot easier. It is a hybrid of the ODS PRINTER model and ODS MARKUP. SAS users can use 
TAGSETS.RTF to more finely control their RTF output. Some examples of what can be accomplished using the 
RTF tagset are: paragraph spacing, column header rotation, paneling, and repeated headers. 
 
One obvious difference between the PRINTER TAGSET / RTF destination and the RTF tagset is that it is not 
necessary to specify BODYTITLE on the RTF call to get your titles and footnotes in the “body” of the text. The 
RTF tagset places titles and footnotes as tables in the RTF document and not as RTF instructions to be placed 
by Microsoft Word. Similarly, SAS and the RTF tagset now determine where the page breaks are for long 
tables and not Word. This means that: larger tables can be rendered since only a page of data and its 

structures have to be held in memory at any one time; procedures like PROC TABULATE can recover row 
headers over implicit page breaks and inform that the table continues on the next page; and merging the RTF 
tagset docs into a master document will not replace the titles and footers from the master. Essentially, SAS 
(and by extension the SAS user) controls RTF output instead of allowing Microsoft Word to make the 
decisions. Using the flexibility of the new RTF (and other) TAGSET(S) allows the creation of custom tagsets 
that fit users’ needs, including the ability to write style functions. The RTF tagset is simply a PROC TEMPLATE 
job that can be modified or inherited from. 
 
There are a number of different resources available for finding out more about TAGSETS.RTF but one of the 
easiest sources of information is in the ODS TAGSETS.RTF call. The RTF tagset has an OPTIONS statement, 
which is also used to rotate column headers. To find out more about the options available, include doc= 
“help” in your OPTIONS statement (enclosed by parentheses) as shown in the code snippet below. This 
syntax also works for the HTML TAGSET. You must specify ODS TAGSETS.RTF (or HTML) or ODS MARKUP 
TAGSET=TAGSETS.RTF as opposed to the RTF destination. The text associated with the TAGSET help file will 
be printed in your log. 
 
The two specific RTF tagset OPTIONS used to rotate column headers are TRHDR and TROWHDRCELL. TRHDR 
allows you to feed in RTF commands relating to the table row header. In this case, we are giving the RTF 
tagset information about the vertical height of the table row header, \trrh9999. The height is measured in 
twips; 20 twips to the point, 72 points to the inch. You will want to play around with the number to get the 
height you need to accommodate your column headers. For the example shown, I used 2750 twips. The 
TROWHDRCELL is where the action occurs. We feed in the RTF command to produce a vertical label (\cltxbtlr) 
in the table row header cells.  

Rotating headers with PROC TABULATE and TAGSETS.RTF 

/* note trrh9999 may need to be adjusted based on label length for var variables */ 
  ods tagsets.rtf file="final_weighted_&basename..rtf" 
  OPTIONS(TRHDR="\trrh2750" TROWHDRCELL="\cltxbtlr" doc="help")         
  style=styles.minimal; 
 

  ods escapechar='^'; 
  proc tabulate data=anal. . . 
  var counter &colvar1 / 
     s={bordercolor=black just=l borderwidth=2 cellwidth=60 font=("Arial",8pt,Normal)}; 

Rotating headers with PROC REPORT and TAGSETS.RTF 

  ods tagsets.rtf file='dummy2.rtf'  
  OPTIONS(TRHDR="\trrh1750" TROWHDRCELL="\cltxbtlr" doc="help") 
  path=odsout style=styles.nmep; 
 
  ods escapechar='^'; 
  proc report nowd data=page1 spacing=8 
        style(report)=[cellpadding=3pt vjust=b] 
        style(header)=[just=center font_face=Arial font_weight=bold    
        font_size=8pt]style(lines)=[just=left font_face=Arial] ls=132; 
     columns rowtitle wgtsum0 printvar1 printvar2 printvar3 printvar4  
             printvar5 printvar6 boldit; 
     define boldit / display ' ' noprint; 
     define rowtitle / style(COLUMN)={just=l font_face=Arial 
                       font_size=8pt cellwidth=200 asis=on} 
                       style(HEADER)={just=l font_face=Arial font_weight=bold 
                       font_size=8pt cellwidth=200 asis=on}; 
     define wgtsum0 / style(COLUMN)={just=c font_face=Arial       
                      background=lightgray font_size=8pt celllwidth=80 asis=on} 
                      style(HEADER)={just=l font_face=Arial font_weight=bold 
                      background=lightgray font_size=8pt celllwidth=80 asis=on}; . . . 

In order to left justify the headers (or bottom justify in this case) it is necessary to specify left justification. 
For the PROC TABULATE example this is done with the VAR statement (highlighted in red in the code) and for 
the PROC REPORT example it is done with the define style header statement.  

It’s important to note that when resetting RTF tagset options (for example STARTPAGE=YES) that you use 
ODS TAGSETS.RTF instead of ODS RTF. In addition, it is a good practice to use the RTF extension for your 
output file instead of the shortcut of using the DOCX extension, as some RTF-specific functionality will be lost 
using the DOCX extension. ODS WORD creates native WORD files which can and should be named with a 
DOCX extension, but there’s less control over the output. 

SAS has some handy “tip sheets” on various topics, including RTF. You can download the RTF related tip 
sheets from SAS’s website. As options can and do change with the tagsets and destinations as improvements 
are made, it’s a good idea to check the documentation prior to using. 

https://support.sas.com/rnd/base/ods/odsrtf/rtf-tips.pdf 
https://support.sas.com/rnd/base/ods/odsrtf/rtf-tagset-tips.pdf 
 
This week’s tip was contributed by Louise Hadden. Louise Hadden is a Lead Programmer Analyst in the Health 
and Environment Division at Abt Associates Inc., specializing in analysis and reporting. 

 

https://support.sas.com/rnd/base/ods/odsrtf/rtf-tips.pdf
https://support.sas.com/rnd/base/ods/odsrtf/rtf-tagset-tips.pdf

	Weekly Tip for Nov. 3, 2020
	Weekly Tip for Nov. 17, 2020
	Weekly Tip for Nov. 24, 2020
	Weekly Tip for Dec. 8, 2020
	Weekly Tip for Dec. 15, 2020
	Weekly Tip for Dec. 22, 2020
	Weekly Tip for Jan. 05, 2021
	Weekly Tip for Jan. 26, 2021
	Weekly Tip for Feb. 2, 2021
	Weekly Tip for Mar. 16, 2021
	Weekly Tip for Mar. 23, 2021
	Weekly Tip for May 4, 2021
	Weekly Tip for May 11, 2021

